
Abstract
For volumerenderingof regulargridsthedisplayof view-planealignedsliceshasproventoyieldbothgoodquality
and performance. In this paperwe demonstrate how to merge the mostimportantextensionsof the original 3D
slicingapproach,namelythepre-integrationtechnique, volumetricclipping, andadvancedlighting. Our approach
allowsthesuppressionofclippingartifactsandachieveshighqualitywhileofferingthe�exibility toexplorevolume
datasetsinteractivelywith arbitrary clip objects.We alsooutlinehowto utilize theproposedvolumetricclipping
approach for the displayof segmenteddata sets.Moreover, we increasethe renderingquality by implementing
ef�cient over-samplingwith the pixel shaderof consumergraphicsaccelerators. We give prove that at least4-
timesover-samplingis neededto reconstructthe ray integral with suf�cient accuracyevenwith pre-integration.
Asanalternativeto thisbrute-forceover-samplingapproach weproposea hardware-acceleratedraycasterwhich
is able to perform over-samplingonly where neededand which is able to gain additional speedby early ray
terminationandspaceleaping.

CR Category: I.3.7 [Computer Graphics]: Three-
DimensionalGraphicsandRealism.

1. Intr oduction

The basicprinciple of volumerenderingwasdescribedby
Kajiya10 as early as in 1984. Since then the availability
of graphicshardware haslead to the establishmentof the
so-called3D slicing method1; 3. The ray integral for each
pixel, which Kajiya reconstructedby ray tracing, is calcu-
latedwith graphicshardwareby renderingview-portaligned
slicesthroughthevolume.In this way theemissionandab-
sorptionalongeachviewing ray throughthevolumeis com-
putedby samplingandblendingthevolumefor eachnumer-
ical integrationstep5. Themainadvantageis that theentire
taskcanbeperformedby thegraphicshardwareandis lim-
itedonly by the�ll rateof thegraphicsaccelerator.

A volume is commonlygiven asa scalarfunction sam-
pled on a uniform grid. The main sourcefor this kind of
volume representationare CT (ComputerTomography) or
MRI (Magnetic ResonanceImaging) scanners,which are
widely usedin medicalimaging.In orderto associateopac-
ity andemissionto thescalarvaluesthevolumedensityopti-
calmodelof Williams etal.25; 13 is usuallyapplied.It de�nes

the opacity and the chromaticityvector to be functionsof
thescalarvalues.Whenslicing a volumein a back-to-front
fashion,the opacityandchromaticityarestoredin an one-
dimensionallook-up table(often referredto asthe transfer
function)which is usedto transformthescalarvalueateach
renderedpixel into an RGBA vector. Thenthe ray integral
canbecalculatedby simplealpha-compositingtheslices.

This basicprincipleof hardware-acceleratedvolumeren-
deringof regulargridshasbeenextendedin severalwaysto
increasethe quality of the renderings.In the following we
give a list of the well establishedextensionsandtheir pur-
poses:

� Ambient and Diffuse Lighting : Theemissionof thevol-
umeis attenuatedby alighting operation22; 24 to givemore
visual cluesregardingthecurvatureof detailsin thevol-
ume.

� Pre-Integration: Theslicing of thevolumeresultsin the
so-calledring artifactswhich aredueto insuf�cient over-
samplingof highfrequenciesin thetransferfunction.This
issueis resolvedby renderingslabsinsteadof slices.The
ray integralof theraysegmentsinsideeachslabis a func-
tion of the scalarvaluesat the entry and exit points of
the ray, andthe thicknessof theslab(seealsoFigure1).
By pre-integrating12; 19 thetransferfunctionfor eachcom-
bination of scalarvaluesthe ray integral can be looked
up easily for eachrenderedpixel using 2D dependent
texturing6.

� Hardware-Accelerated Pre-Integration: Since pre-
integration involves a fair amountof numericalopera-
tions, a changeof the transfer function leadsto a de-
lay of the volume visualization while the 2D depen-
dent texture hasto be recomputed.If self-attenuationis
neglected, interactive updaterates of up to 10 Hertz6

are achieved for a transfer function with 256 entries.
However, this introducesartifacts in regions where the
transferfunction is very opaque.A bettersolution is to
speedup the computationof the exact ray integral by
using graphicshardware18; 7. The quantizationartifacts
that normally arisewith an 8-bit frame buffer are over-
comeby usingthe �oating point rendertargetof modern
graphicsacceleratorssuchas the ATI Radeon97002 or
NVIDIA GeForceFX16. A 256-stephardware-accelerated
pre-integration with full �oating point precision takes
about50millisecondson theATI Radeon9700.

� Material Properties: Besidesthe pre-integration of the
transferfunction,onecanalsopre-integratematerialprop-
ertieswhichdescribethefractionof ambient,diffuse,and
specularlighting whichhasto beappliedto eachrendered
slab14.

� Volumetric Clipping: To exploretheinteriorof avolume
convenientlyvolumetricclipping23 is usedto cutoff parts
of thevolumewhich otherwisewould hide importantin-
formation.For that purposethe clip geometryis de�ned
byaniso-surfaceof anadditionalclip volume(whichneed
not necessarilybeof thesamesizeasthescalarvolume).
Duringrenderingtheopacityof eachpixel is setto zeroin
thepixel shaderif thecorrespondingclip valueexceedsa
certainthreshold.

In summary, the describedextensionshave led to high
qualityvolumerenderingusingthegraphicshardware.How-
ever, the combinationof the pre-integrationtechniquewith
volumetricclipping hasbeenan unsolved problem.Hence
our �rst goal is to demonstratethecombinationof thesead-
vancedtechniques.To increasetheaccuracy evenfurther, we
proposea hardware-acceleratedray casterwhich is ableto
adaptthesamplingfrequency to the reconstructionerrorof
theray integral.

2. Accurate Volumetric Clipping

In orderto combinepre-integrationwith volumetricclipping
basically threestepsare necessary. Let Cf andCb be the
scalarvaluesof theclip volumeattheentryandexit pointsof
theray segmentin therange[0;1]. We alsoassumethat the
volumecorrespondingto clip valuessmallerthan0:5 should
beclippedaway. If bothparametersCf andCb areabove0:5,
thentheslabis completelyvisible andno specialtreatment
is necessary. ConsideringthecaseCf < 0:5 andCb > 0:5 as
depictedin Figure1, only thedarkbluepart of thevolume
hasto berendered.In this casewe �rst have to setthefront
scalarvalueSf to thevalueS0

f atentrypoint into theclipped
region. Thuswe performa look-up into the pre-integration

tablewith theparameters(S0
f ;Sb) ratherthanwith (Sf ;Sb).

In thegeneralcaseSf is replacedby S0
f accordingto

r =
�

[0:5� Cf ]
Cb � Cf

�
; S0

f = (1� r)Sf + rSb :

The brackets denoteclamping to the range[0;1]. For the
caseCf > 0:5 andCb < 0:5, thefront scalarvalueneednot
beadjusted,whichis expressedby r = 0.Thesameholdsfor
thecaseCf > 0:5 andCb > 0:5. Similarily, theparameterSb
is replacedby S0

b asfollows:

g = 1�
�

[0:5� Cb]
Cf � Cb

�
; S0

b = (1� g)Sf + gSb :

If bothclip valuesarebelow 0:5, theray segmentis clipped
entirely, andthusthescalarvaluesdonotmatter.

Sf Sb

l

Slab

Sf

Figure 1: Using slabs insteadof slices for pre-integrated
volumerenderingasintroducedby Engel et al.6. Thescalar
valuesat theentryandtheexit point of theviewing ray are
denotedbySf andSb, respectively. Thethicknessof theslab
is denotedby l . Thedarkblueregion remainsafter volumet-
ric clipping. For thispurpose, Sf hasto bereplacedbyS0

f .

Thesecondproblemwe have to careaboutis thereduced
length l0 of the clipped ray segment.The numericalpre-
integrationdependsonthethreeparametersSf , Sb andl (see
Figure1). Using the optical modelof Williams andMax25

with the chromaticityvector k and the scalaroptical den-
sity r , theray integral for eachray segmentis givenasfol-
lows:

Sl (x) = Sf +
x
l
(Sb � Sf )

C(Sf ;Sb; l ) =
Z l

0
e�

Rt
0 r (Sl (u))duk(Sl (t)) r (Sl (t))dt

q(Sf ;Sb; l ) = e�
Rl

0 r (Sl (t))dt ; a = 1� q

Now we have to distinguishbetweentwo differenttypesof
transferfunctions.For atransferfunctionthatde�nesasetof
isosurfacesthereducedraysegmentlengthl 0hasnoeffecton
the integratedchromaticityC andthe integratedopacitya.
For the other commoncaseof a transferfunction de�ned
by a lookuptablethethree-dimensionalintegrationproblem
canbereducedto two dimensions.

Thepre-integrationis performedfor theconstantrayseg-
mentlengthl . Let thevisible fractionof theslabbedenoted
by b = l0=l . Thenthetransparency q0of theclippedrayseg-
mentis the pre-integratedtransparency q raisedto the b-th
powerbecause

Z l0

0
r (S0

l (t
0))dt0= b

Z l

0
r (Sl (t))dt ;

q0= e� b
Rl

0 r (Sl (t))dt =
�

e�
Rl

0 r (Sl (t))dt
� b

= qb :

In practicehowever, a �rst orderapproximationis suf�cient
if thethicknessof theslabsis reasonablesmall.

If self-attenuationis neglected,theemissionof theclipped
ray segmentis given by C0 = bC. The computationof the
correct emissioncannotbe expressedin terms of a two-
dimensionalpre-integration. For accurateresultsa three-
dimensionalpre-integration table hasto be used,but high
orderpolynomialapproximations7 doaremarkablygoodjob
to get rid of the hugethree-dimensionalpre-integrationta-
ble.Asidefrom thatissue,theneglectionof self-attenuation
is oftenaverygoodassumptionin practice.

Thethird andlastproblemis thelighting of theslab. Sam-
pling the gradientat a single point results in severe ring
artifacts23. A bettersolutionis to samplethegradientat the
entryandexit pointsandto adjustthegradientsin thesame
fashionasthescalarvalues.Thenthe�nal light intensityof
theslabis theaverageof bothadjustedlight intensities.The
operationsinvolvedherecanbeexpressedasa linear inter-
polationof thegradients.We denotetheinterpolationfactor
by a.

Insteadof calculatingthefactorsfor theadjustmentof the
scalarvalues,theemission,theopacity, andthegradientsin
the pixel shaderwe pre-computethe factorsfor all combi-
nationsof theclip valuesCf andCb andstorethemin a 2D
dependenttexture.Thecontentof eachR/G/B/A channelof
thedependenttextureasdepictedin Figure2 correspondsto
theadjustmentfactorswith thesamename.

Figure 2: 2D dependenttexture containing the adjust-
mentfactorsusedfor accuratevolumetricclipping(R/G/B/A
channelsaredepictedfromleft to right).

Using the pixel shaderversion2.015, both scalarvalues
canbeadjustedby asinglelinearinterpolation("lrp" instruc-
tion). The entire pixel shaderperforming pre-integration,
ambientanddiffuselighting, andaccuratevolumetricclip-
ping is givenin Appendix8.2.

A comparisonof the renderingquality betweenour ap-
proachanda naive applicationof the methodby Weiskopf
et al.23 to pre-integration is depictedin Figure 3 (seeAp-
pendix8.1and8.2for thepixel shaders).A spherehasbeen
cutoutof theBucky Ball datasetsothattheholesof thecar-
bonringsarevisible asgreenspots.Both imageshave been
renderedwith only 32 slices.Whereastheslicesareclearly
visible on the left, our methodreproducesthe clippedvol-
umeaccurately. This meansthatour methodis alsoexact in
theviewing direction.

Figure 3: Comparisonof volumetricclipping quality. Left:
Naiveapproach. Right: Accuratemethod.

The presentedvolumetric clipping algorithm can also
be extendedto rendersegmentedvolumes21. In conjunc-
tion with thepre-integrationtechniquethis wasanunsolved
problemaswell. If eachsegmentis de�nedasaclip volume,
thisproblemcanbereducedto volumetricclipping.

For the caseof two segmentedvolumesa clip volumeis
setupto includeoneentiresegment.For thesecondsegment
weuseanadditionaltransferfunction.For thepartof theslab
whichis clippedawayweperformanotherdependenttexture
lookup into the secondpre-integration tableandblend the
two partial slabsdependingon their orientation.For each
additionalsegmentthisprocedurehasto berepeated.

3. Over-Sampling

In the last sectionwe have describedthe combinationof
the pre-integration techniquewith volumetric clipping to
improve thequality of volumevisualizations.Nevertheless,
even the mentionedadvancedtechniquesare no guarantee
for artifact-freerenderings.

A fundamentalassumptionof volumerenderingis thatthe
scalarvaluesof thevolumeareinterpolatedtri-linearly. The
pre-integrationtechniquehaseliminatedmany of therender-
ing artifactsthatarisefrom slicing thevolume(comparetop
right andbottom left imageof Figure4), but it assumesa
linearprogressionof thescalarvaluesinsidetheslabs.This
assumptiondoesnotmatchtheactualcubicbehaviour of the
scalarvalues.Furthermore,the scalarfunction may have a
sharpbendif avoxel boundaryis crossedinsidea slab. This

non-lineareffectis ampli�ed by lighting in regionswith high
secondderivatives.

Consideringpre-integration alone the exact reconstruc-
tion of the ray integral is a four-dimensionalproblem(for
constantray segmentlengthl ). Includingtheeffect of light-
ing it canonly besolvedef�ciently with over-sampling.The
top row of Figure4 illustratesthe artifactsthat may occur
without over-sampling.Using 2-timesover-samplinghelps
a lot, but thereis still a largedifferencewhensteppingup to
4-times over-sampling.Fortunately8-times over-sampling
leadsto insigni�cant improvementsonly, so that 4-times
over-samplingis alreadya goodchoicefor highestquality
in practice.

Current graphicshardware allows to perform multiple
blendingstepsat oncein the pixel shader. We have imple-
menteda pixel shaderprogramthatblendsfour slabsinter-
nally beforeit writes the result back into the framebuffer
(seeAppendix8.3).As a sideeffect, thefour slabs(seeFig-
ure 1) are blendedwith the high internal precisionof the
pixel shader.

For awindow sizeof 5122 pixelstherenderingtimeis be-
tween1 and1.7secondsfor thismulti-steppedapproach.Be-
causeof the increasedcachecoherencewe achieve a speed
up of 70%-120%over the single-steppedversiondepend-
ing on the dataset and viewing parameters.We conclude
that2-timesover-samplingcomesalmostfor freewhenus-
ing multi-stepping.For very high quality requirementsone
hasto acceptabouttwice therenderingtimeasnormal.

4. Hardware-AcceleratedRay Casting

The previously discussedover-sampling approach sup-
pressesartifacts causedby neglecting tri-linear interpola-
tion, thecrossingof voxel boundaries,andthenon-linearbe-
haviour of lighting. While this brute-forceapproachis prac-
tically artifact-free,it doesnotexploit any spatialcoherence
in thedatasetto increasetherenderingperformance.In com-
parisonto brute-forceover-sampling,a ray caster4; 11 is able
to adaptthe samplingrate to the actualinformation in the
dataset.This oftenleadsto a drasticallyreducednumberof
samplepoints.

Just like the �e xibility of currentgraphicshardware al-
lowshardware-acceleratedraytracing17 it alsoallowsto per-
form ray castingcompletelyin hardware. In the following
we describea smarthardware-acceleratedray casterwhich
appliesan error-driven samplingschemeand additionally
performsearlyray termination.

Thepre-integrationtechniqueisbasedonapiece-wiselin-
earapproximationof the ray integral, thusall higherorder
frequenciesare neglected.In the optimal caseone would
choosethe step length suchthat the differenceof the ex-
actray integral andthelinearapproximationis lower thana
pre-de�nedthreshold.But this solutionis infeasiblesinceit
requiresthepre-integrationof theentiredataset.

Figure 4: A leaveof the Bonsai20. Thedegradationof ren-
dering quality is displayedfrom top left to bottom right:
with pre-integration and 4-timesover-sampling, with pre-
integration and2-timesover-sampling, with pre-integration
andwithoutover-sampling, andneitherwith pre-integration
nor over-sampling. In thebottomright cornerof each image
thezoomof a critical region is depictedwhich shouldshow
a smoothcolor transition.Dueto slicingartifactsin thebot-
tom right image arti�cial bandsare visible. Theseremain
evenwith 2-timesoversamplingbut almostdisappearwith
4-timesoversampling.

Insteadof solving the global problemwe resortto solv-
ing the local problem,which basicallyrequiresthe compu-
tation of the local approximationerror of the scalarfunc-
tion andthe visibility of this error, which is determinedby
thetransferfunction.A naturalstrategy to computethestep
lengthshouldbebasedonatleastthesecondderivativeof the
scalarfunctionandthepre-integratedemissionandopacity
of thetransferfunction.Thisapproach,whichwecall adap-
tivepre-integration, automaticallysubsumesthewell-known
accelerationtechniqueof spaceleaping5.

Conceptually, theprocessof raycastingcanbedividedin
threemaintasks,whicharetheraysetup,thesamplingof the
ray, andthenumericalintegrationof thesamples.In orderto
exploit themassiveparallelismof thegraphicshardware,we
computetheintegrationfor all raysin parallel.For eachstep
the front facesof theboundingbox arerenderedto process
all visible raysin a front-to-backfashion(seeFigure5).

Theray positionis determinedby a �oating point render
targetwhich containsthecurrentray parameter. For a given
samplingdistancel therayparameteris updatedaccordingly

andwrittenbackinto the�oating point rendertarget.Simul-
taneously, the integratedchromaticityandopacityarewrit-
ten to additional�oating point rendertargets(we usetwo
targetfor theRGandBG channels).A so-calledimportance
volumeis usedto extract the maximumisotropicsampling
distanceaspreviouslyde�ned.

Bounding Box

Screen

already terminated

actual position

initial position

ray parameter

Figure 5: Ray castingscheme:All viewing rays are pro-
cessedsimultaneously. For each integration step the pre-
integration techniqueis used.After each stepthecolors are
blendedand the ray parametercorrespondingto the next
samplingpositionis writtenback into alternatingrendertar-
gets(ping-pongrendering).

A ray is terminatedif it either leaves the volume or if
the integratedopacity is approximatelyone.If a ray is ter-
minatedwe have to prevent further computation.For this
purpose,we usethe Z-buffer since it is testedbeforeany
pixel shadercomputation.Thisalsoincludestexturereads.If
theray endswe settheZ-buffer suchthatthecorresponding
pixel is not processedanymore.This requiresanadditional
renderingpass.

To �nd out whetherall rays have beenterminatedwe
apply an asynchronousocclusion query. In total, we re-
quire 2n � 1 passeswith n beingthe maximumnumberof
samplesfor the raysin the generatedimage.The �rst pass
is additionallysettingup theinitial ray parameters.In order
to allow re-implementationof theraycasterthepixel shader
2.0codeis includedin theAppendices8.4,8.5,and8.6.

In contrastto the brute-forceslicing approachthe total
numberof samplesdependson thetransferfunctionandon
thecoherenceof thevolumedata.Also consideringearlyray
termination,thenumberof samples(seeFigure6) is always
lessthanfor the slicing approach,but the samereconstruc-
tion quality is achieved.In additionto this,all computations
including blending are performedwith full �oating point
precision,so that artifactsdueto framebuffer quantization
cannotoccur(compareFigure7).

5. Results

All performancemeasurementshave beenconductedon a
PC equippedwith an ATI Radeon9700graphicsaccelera-

Figure 6: Numberof samplingstepsfor different transfer
functions:On the left theoriginal image is depictedandat
the center the correspondingnumberof samplingstepsis
shown(Whitecorrespondsto 512 samples).On the right a
more opaquetransferfunctionwaschosento illustrate the
impactof early ray termination.

Figure 7: Quality comparisonbetweenslicing with pre-
integration and 4-timesover-sampling(left) and ray cast-
ing with full �oating point accuracy and adaptive pre-
integration (right). On the left highly transparentareasare
neglecteddueto 8 bit framebuffer quantization.

tor. Figure8 showsaBonsai20 of size2563 with andwithout
accuratevolumetric clipping using 4-timesover-sampling.
Renderingtimes are approximately2 and 1.5 seconds,re-
spectively. The renderingtimes for the ray-castedbonsai
with opaqueandsemi-transparenttransferfunctionsarebe-
tween1 and 3 secondsper frame (Figures6 and 9). The
smallerNegHIP dataset(Figure7) with a sizeof 1283 vox-
els took 2.2 secondswith a semi-transparenttransferfunc-
tion and0.7secondswith anopaquetransferfunction.

In comparisonto traditionalslicingthequalityof raycast-
ing is alwayshigher. Dependingon thetransferfunctionray
castingmayevenbefaster. By increasingtheerrorthreshold,
therenderingtime caneasilybereducedto 0.1 secondsper
frame.The artifacts introducedhereare much lessvisible
thanfor theanaloguecaseof reducingthenumberof slices.

With the upcomingof graphicsacceleratorssuchas the
NVIDIA GeForce FX, we expect the performanceof a
hardware-acceleratedray casterto increasemuchmorethan
the performanceof regular multi-steppedslicing. The rea-
sonsare as follows: First, an additionalpassis no longer

Figure8: Bonsaiwith andwithoutvolumetricclipping.

Figure9: Hardware-acceleratedraycasting.

neededfor ray termination.Secondly, multiple stepscanbe
performedatoncebecauseof ahighermaximuminstruction
count.As buffer writesbetweenthepassesconsumemostof
the renderingtime, this overheadcan be reduceddramati-
cally with theNVIDIA GeForceFX.

The performanceof a software ray caster11 is approxi-
matelythe sameasfor our hardware-acceleratedapproach.
However, the software ray casterachieves its bestperfor-
manceonly without pre-integration,sotheresultingquality
is not comparable.

6. Conclusion

We have combined volumetric clipping with the pre-
integration techniqueto achieve high-qualityvolumevisu-
alizations.Sincepre-integrationdoesnot completelysolve
theproblemof thereconstructionof theray integral,wepre-
senteda hardware-acceleratedray caster. By adaptive pre-
integrationof the viewing raysthe reconstructionerror can
beguaranteedto bebelow thethresholdcontrastsensitivity9.
Theraycasteralsoappliespre-integration,spaceleapingand
early ray termination.In the futurewe plan to speedup the
ray casterby usinghierarchicalapproaches8 thatwill allow
thevisualizationof very largedatasets.

8. Appendix

8.1. NaiveVolumetric Clipping:
ps_2_0 // ps 2.0
dcl_volume s0 // 3D RGBA volume
dcl_2d s1 // 2D pre-integration table (RGBA)
dcl_volume s2 // 3D clip volume (Luminance)
dcl t0.xyz // texcoord front
dcl t1.xyz // texcoord back
dcl t2.xyz // normalized eye vector
def c0,0.5,1.0,2.0,0.0 // constant definitions
texld r0,t0,s0 // get front scalar value
texld r1,t1,s0 // get back scalar value
texld r4,t0,s2 // get front clip value
mov r2.r,r0.a // move front scalar value to r2.r
mov r2.g,r1.a // move back scalar value to r2.g
texld r2,r2,s1 // pre-integration lookup
mul r2.rgb,r2,r2.a // post-multiply alpha
mad r0.rgb,c0.b,r0,-c0.g // expand normal
dp3_sat r3,t2,r0 // calculate light intensity
mul r2.rgb,r2,r3 // multiply emission with intensity
add r4.r,r4.r,-c0.r // depending on comparison with 0.5..
cmp r2.a,r4.r,r2.a,c0.a // ..set opacity to zero
mov oC0,r2 // move to output register

8.2. Accurate Volumetric Clipping:
ps_2_0 // ps 2.0
dcl_volume s0 // 3D volume
dcl_2d s1 // 2D pre-integration table
dcl_volume s2 // 3D clip volume
dcl_2d s3 // clip coefficient table (RGBA)
dcl t0.xyz // texcoord front
dcl t1.xyz // texcoord back
dcl t2.xyz // normalized eye vector
def c0,0.5,1.0,2.0,0.0 // constant definitions
texld r0,t0,s0 // get front scalar value
texld r1,t1,s0 // get back scalar value
texld r4,t0,s2 // get front clip value
texld r5,t1,s2 // get back clip value
mov r4.g,r5.r // move back clip value to r4.g
texld r4,r4,s3 // get clip coefficients
lrp r2,r4,r1.a,r0.a // adjust scalar values
texld r2,r2,s1 // pre-integration lookup
mul r2.rgb,r2,r2.a // post-multiply alpha
mad r0.rgb,c0.b,r0,-c0.g // expand front normal
dp3_sat r3,t2,r0 // calculate front light intensity
mad r1.rgb,c0.b,r1,-c0.g // expand back normal
dp3_sat r3.g,t2,r1 // calculate back light intensity
add_sat r3,c0.r,r3 // bias light intensities
lrp r5.r,r4.a,r3.g,r3.r // adjust light intensities
mul r2.rgb,r2,r5.r // multiply emission with intensity
mul r2,r2,r4.b // attenuate emission and opacity
mov oC0,r2 // move to output register

8.3. 4Steps@Once:
ps_2_0 // ps 2.0
dcl_volume s0 // 3D volume
dcl_2d s1 // 2D pre-integration table
dcl t0.xyz // texcoord front
dcl t1.xyz // texcoord back
dcl t2.xyz // normalized eye vector
dcl t3.xyz // second texcoord front
dcl t4.xyz // second texcoord back
dcl t5.xyz // central texcoord

def c0,0.5,1.0,2.0,0.0 // constant definitions
texld r0,t0,s0 // get front scalar value
texld r1,t1,s0 // get back scalar value
texld r3,t3,s0 // get second front scalar value
texld r4,t4,s0 // get second back scalar value
texld r6,t5,s0 // get central scalar value
mad r0.rgb,c0.b,r0,-c0.g // expand front normal
dp3_sat r0.r,t2,r0 // calculate front light intensity
add_sat r0.r,c0.r,r0.r // bias light intensity
mad r3.rgb,c0.b,r3,-c0.g // expand second front normal
dp3_sat r3.r,t2,r3 // calculate front light intensity
add_sat r3.r,c0.r,r3.r // bias light intensity
mad r6.rgb,c0.b,r6,-c0.g // expand central normal
dp3_sat r6.r,t2,r6 // calculate front light intensity
add_sat r6.r,c0.r,r6.r // bias light intensity
mad r4.rgb,c0.b,r4,-c0.g // expand second back normal
dp3_sat r4.r,t2,r4 // calculate front light intensity
add_sat r4.r,c0.r,r4.r // bias light intensity
mov r2.r,r4.a // move second back scalar value
mov r2.g,r1.a // move back scalar value
texld r2,r2,s1 // pre-integration lookup
mul r2.rgb,r2,r2.a // post-multiply alpha
add r2.a,c0.g,-r2.a // compute transparency
mul r2.rgb,r2,r4.r // multiply emission with intensity
mov r5.r,r6.a // move central scalar value
mov r5.g,r4.a // move second back scalar value
texld r5,r5,s1 // pre-integration lookup
mul r5.rgb,r5,r5.a // post-multiply alpha
add r5.a,c0.g,-r5.a // compute transparency
mul r5.rgb,r5,r6.r // multiply emission with intensity
mad r2.rgb,r2,r5.a,r5 // blend operation (1st)
mul r2.a,r2.a,r5.a // accumulate alpha
mov r5.r,r3.a // move second front value
mov r5.g,r6.a // move central scalar value
texld r5,r5,s1 // pre-integration lookup
mul r5.rgb,r5,r5.a // post-multiply alpha
add r5.a,c0.g,-r5.a // compute transparency
mul r5.rgb,r5,r3.r // multiply emission with intensity
mad r2.rgb,r2,r5.a,r5 // blend operation (2nd)
mul r2.a,r2.a,r5.a // accumulate alpha
mov r5.r,r0.a // move front scalar value
mov r5.g,r3.a // move second front scalar value
texld r5,r5,s1 // pre-integration lookup
mul r5.rgb,r5,r5.a // post-multiply alpha
add r5.a,c0.g,-r5.a // compute transparency
mul r5.rgb,r5,r0.r // multiply emission with intensity
mad r2.rgb,r2,r5.a,r5 // blend operation (3rd)
mul r2.a,r2.a,r5.a // accumulate alpha
add r2.a,c0.g,-r2.a // compute opacity
mov oC0,r2 // move to output register

8.4. Ray Casting: Ray Setupand First Integration Step

ps_2_0 // ps 2.0
def c0,0.0,2.0,63.75,1.0 // constant definitions
def c1,1.0,0.0,0.333,0.5 // (c2,c3) is the bounding box
dcl t0.xyz // first intersection point with volume
dcl t1.xyz // ray direction (not normalized)
dcl t2.xyz // scaling of texture coordinates
dcl t3.xyz // direction to light source
dcl_volume s0 // 3D volume
dcl_volume s1 // 3D table holding sampling distances
dcl_volume s2 // 3D pre-integration table
texld r2,t0,s0 // sample volume at first intersection
texld r5,t0,s1 // get distance to next sampling point
nrm r0.xyz,t1 // normalize ray direction..
mul r0.xyz,r0,t2 // ..and multiply with scaling factors
add r4.xyz,c2,-t0 // calculate signed distances..
add r3.xyz,c3,-t0 // ..to the bounding box
rcp r6.x,r0.x // calculate reciprocal..
rcp r6.y,r0.y // ..of each component..
rcp r6.z,r0.z // ..of ray direction
mul r3.xyz,r3,r6 // calculate distance to bounding box..
mul r4.xyz,r4,r6 // ..in ray direction
max r6.xyz,r3,r4 // get closest..
min r4.x,r6.x,r6.y // ..non-negative..
min r1.y,r4.x,r6.z // ..intersection point
mov r1.xzw,c0.x // store intersection 4 ray termination
mad r1.xz,r5.x,c1,r1.x // compute new sampling position

min r1.x,r1.x,r1.y // clamp to volume boundary
mad r3.xyz,r1.x,r0,t0 // calculate second sampling point..
texld r4,r3,s0 // ..and sample volume at this position
mad r2.xyz,r2,c0.y,-c0.w // extract gradient
nrm r8.xyz,r2 // normalize gradient
add r6.w,r1.x,-r1.z // get interval length of current step
lrp r6.xy,c1,r2.w,r4.w // setup texcoords 4 pre-integration..
mad r6.z,r6.w,c1.z,-c1.z // ..including ray segment length l
texld r7,r6,s2 // pre-integration lookup
dp3_sat r8.w,r8,t3 // calculate light intensity
mul r6.x,c0.z,r6.w // change opacity to correctly..
add r7.a,c0.w,-r7.a // ..represent the ray segment length..
pow r10.a,r7.a,r6.x // ..which includes raising it..
add r7.a,c0.w,-r10.a // ..to the power of l
mul r7.rgb,c1.w,r7 // bias color 4 lighting
mad r7.rgb,r8.w,r7,r7 // multiply emission with intensity
mul r7.rgb,r7,r7.a // post-multiply alpha
mov r0,r7.b // split color 4 floating point output..
mov r0.g,r7.a // ..into two render targets
mov oC0,r7 // output color (RG)
mov oC1,r0 // output color (BA)
mov oC2,r1 // output ray parameter

8.5. Ray Casting: Ray Termination

ps_2_0 // ps 2.0
def c0,-1.0,-1.0,0.0,0.0 // constant definitions
def c1,0.996,0.0,0.0,0.0 // threshold equals 1/256
dcl t4.xyzw // position of pixel in screen space
dcl_2d s3 // previous ping-pong buffers holding..
dcl_2d s4 // ..floating point colors
dcl_2d s5 // buffer containing sample positions
texldp r1,t4,s5 // get sampling position..
mov r1.zw,c0.z // ..and clear unused part
texldp r8,t4,s4 // get opacity
add r0,r1.x,-r1.y // calc distance to volume boundary
cmp r1,r0,c0,r1 // check if distance equals 0
add r0,c1.x,-r8.g // compare opacity against threshold..
cmp r1,r0,r1,c0 // ..4 early ray termination
mov r0,-r1 // leave shader if the ray..
texkill r0 // ..is not being terminated
texldp r7,t4,s3 // get remaining parts of color
mov oC0,r7 // output floating point color into..
mov oC1,r8 // ..alternate ping-pong buffers

8.6. Ray Casting: OneAdditional Integration Step

ps_2_0 // ps 2.0
def c0,0.0,2.0,63.75,1.0 // constant definitions
def c1,1.0,0.0,0.333,0.5 // (c2,c3) is the bounding box
dcl t0.xyz // first intersection point with volume
dcl t1.xyz // ray direction (not normalized)
dcl t2.xyz // scaling of texture coordinates
dcl t3.xyz // direction to light source
dcl t4.xyzw // position of pixel in screen space
dcl_volume s0 // 3D volume
dcl_volume s1 // 3D table holding sampling distances
dcl_volume s2 // 3D pre-integration table
dcl_2d s3 // previous ping-pong buffers holding..
dcl_2d s4 // ..floating point colors
dcl_2d s5 // buffer containing sample positions
texldp r1,t4,s5 // get sampling position
texldp r7,t4,s3 // get color from two..
texldp r8,t4,s4 // ..floating point render targets
mov r7.z,r8.x // combine the color channels of..
mov r7.w,r8.y // ..the two render targets
nrm r0.xyz,t1 // normalize ray direction..
mul r0.xyz,r0,t2 // ..and multiply with scaling factors
mad r3.xyz,r1.x,r0,t0 // calculate first sampling point
texld r2,r3,s0 // sample volume at first sampling point
texld r5,r3,s1 // get distance to next sampling point
mad r1.xz,r5.x,c1,r1.x // compute new sampling position
min r1.x,r1.x,r1.y // clamp to volume boundary
mad r3.xyz,r1.x,r0,t0 // calculate second sampling point..
texld r4,r3,s0 // ..and sample volume at this position
mad r2.xyz,r2,c0.y,-c0.w // extract gradient
nrm r8.xyz,r2 // normalize gradient
add r6.w,r1.x,-r1.z // get interval length of current step

lrp r6.xy,c1,r2.w,r4.w // setup texcoords 4 pre-integration..
mad r6.z,r6.w,c1.z,-c1.z // ..including ray segment length l
texld r9,r6,s2 // pre-integration lookup
dp3_sat r8.w,r8,t3 // calculate light intensity
mul r6.x,c0.z,r6.w // change opacity to correctly..
add r9.a,c0.w,-r9.a // ..represent the ray segment length..
pow r10.a,r9.a,r6.x // ..which includes raising it..
add r9.a,c0.w,-r10.a // ..to the power of l
mul r9.rgb,c1.w,r9 // bias color 4 lighting
mad r9.rgb,r8.w,r9,r9 // multiply emission with intensity
mul r9.rgb,r9,r9.a // post-multiply alpha
add r8.a,c0.w,-r7.a // blend new color..
mad r7,r9,r8.a,r7 // ..with the old one
mov r0,r7.b // split color 4 floating point output..
mov r0.g,r7.a // ..into two render targets
mov oC0,r7 // output color (RG)
mov oC1,r0 // output color (BA)
mov oC2,r1 // output ray position

Top left: Naivevolumeclippingapproach(renderedBucky Ball showsslicingartifacts).Top right : Accuratevolumeclipping
method.Bottom: Hardware-acceleratedraycasting(Bonsairenderedwith ATI Radeon9700).

